# Maintainer: Jan Alexander Steffens (heftig) <heftig@archlinux.org>

pkgname=dconf-editor
pkgver=43.0
pkgrel=1
pkgdesc="GSettings editor for GNOME"
url="https://wiki.gnome.org/Apps/DconfEditor"
arch=(x86_64)
license=(GPL3)
groups=(gnome-extra)
depends=(libhandy dconf)
makedepends=(vala meson git)
options=(debug)
_commit=27b4f9d1d53783507f03eb9b6a140f0646ce746e  # tags/43.0^0
source=("git+https://gitlab.gnome.org/GNOME/dconf-editor.git#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/[^-]*-g/r&/;s/-/+/g'
}

prepare() {
  cd $pkgname
}

build() {
  arch-meson $pkgname build
  meson compile -C build
}

check() {
  meson test -C build --print-errorlogs
}

package() {
  meson install -C build --destdir "$pkgdir"
}

# vim:set sw=2 sts=-1 et:
